AAEC/TM437

AAEC/TM437

AUSTRALIAN ATOMIC ENERGY COMMISSION

RESEARCH ESTABLISHMENT
LUCAS HEIGHTS

STUDIES IN COMPUTER COUPLED RADIOCHEMICAL ANALYSIS - PART 1.
PEAK LOCATION AND PEAK ENERGY MEASUREMENT IN SCINTILLATION SPECTROMETRY

by

A.R. PALMER



February 1968

AUSTRALIAN ATOMIC ENERGY COMMISSION

RESEARCH ESTABLISHMENT

LUCAS HEIGHTS

STUDIES IN COMPUTER COUPLED RADIOCHEMICAL ANALYSIS - PART 1

PEAK LOCATION AND PEAK ENERGY MEASUREMENT IN SCINTILLATION SPECTROMETRY

by

A. R. PALMER

ABSTRACT

A FORTRAN IV (Level H) programme is described which locates photopeaks in the output of a multichannel analyser scintillation spectrometer assembly. The programme is particularly useful for processing the output of large capacity analysers working in conjunction with a high resolution Ge(Li) detector.

CONTENTS

1. INTRODUCTION

In recent years there has been considerable interest in the subject of computer coupled activation analysis. The logical result of coupling a computer to activation analysis is seen in the present generation of analysers where data are stored directly in a small computer rather than in a conventional multi-channel pulse height analyser. Whilst the engineering aspects of such assemblies are now well understood, there is much to be done in the field of data processing.

Peak location by computer was used for a while by Drew et al. (1962) and Kuykendall and Wainerdi (1960), and it was discussed by Yule (1966). Although the trend has been to supplement peak location routines by more sophisticated mathematical treatments, such routines are still very useful in activation analysis studies, particularly since the advent of high resolution Ge(Li) detectors. These require the use of many channels if full advantage is to be taken of their inherently good resolution. Such detector-analyser combinations generate large quantities of data which take a long time to examine manually.

The work reported here is a FORTRAN IV Level H programme developed for peak location and peak energy measurement in gamma scintillation spectrometry. It is intended primarily for use with the output from large capacity multi-channel analysers connected to a high-resolution detector. As its prime purpose is peak energy measurement, it is used largely for deciding which isotopes are present in a sample. Subsequent reports will deal with the problems of energy assignment, evaluation of peak areas and spectrum stripping.

2. FACILITIES AVAILABLE

The facilities available to Chemistry Division staff at the A.A.E.C. comprise a 1024-channel Nuclear Data Pulse Height Analyser (Model FMR) and a range of detectors consisting of NaI (Tl) crystals and a Li(Ge) solid state detector. Output from the detector-analyser is available either as printed lines via an electric typewriter or as punched tape from a Nuclear Data Model Punch-Reader unit. At present, to process data with the site computer (IBM-360) it is necessary to read punched tape from the analyser with the computer and produce a set of cards or magnetic tape for subsequent use as programme input. This step, although time-consuming, has proved quite satisfactory in practice. Punched card data storage has so far been preferred since it permits ready insertion of title cards and other data required for execution of the computer programme. The programme described here is thus designed solely for card input.

3. BASIS OF PROGRAMME

The first step in processing of the input data is to sort and reproduce the input in a more readable form. This is a valuable feature of the programme since the output from the IBM typewriter is in groups of eight channels without channel identification. Subroutine SORT reproduces the input data arranging it with 10 channels per line, 400 channels per page with an initial channel identification for each line.

For subsequent operations, data convolution is applied. This technique, first applied in analytical chemistry by Savitzky and Golay (1964) and then applied to scintillation spectrometry by Blackburn (1965), largely eliminates statistical scatter in the data and produces a smoothed spectrum. Subroutine SMOOTH convolutes the raw data and the choice of 5, 7 or 9 point cubic smoothing is available corresponding to the value of NOPTS supplied by the user. The smoothed data are printed out in the same format as that used for the input data.

After data convolution, photopeaks are located by means of the subroutine SEARCH. The following techniques are applied sequentially to locate photopeaks:

(i) The smoothed first derivative (D) is examined to determine at which channel (n) it changes sign. Next, the spectral region either side of this point is examined in more detail. The computer looks for groups of channels which satisfy the following criteria:

$$D_{n-1} > 0 \quad ; \quad D_n \leq 0 \quad ; \quad D_{n+1} < 0$$

(ii) The computer now examines the spectrum over the channels n-WIDTH to n+WIDTH and counts the number (n-) of channels for which $D \leq 0$ between channels n+1 to n+WIDTH and the number of channels (n+) for which $D \geq 0$ between channels n-1 to n-WIDTH. As programme input, the experimenter provides another parameter FWIDTH. If both n- and n+ are > FWIDTH, a peak is considered to exist near channel n and the value of n is stored for further processing. This test verifies that the spectrum does rise to a maximum in the vicinity of channel n. The value of the parameter WIDTH is supplied as input by the programme user.

(iii) Since peaks satisfying the preceding criteria can be produced by chance, the peak is examined to determine if it is statistically valid (Drew et al. 1962, Kuykendall and Wainerdi 1960). The test for statistical significance at channel n is based upon the following assumptions:

(a) The number of counts in each channel is a random sampling of a normal distribution.

(b) There is a chance that a channel can contain more counts than several channels on either side, and yet not actually be a peak.

(c) Comparison of the count in channel n with the count in channels $n + \Delta n$ and $n - \Delta n$ will indicate the existence or non-existence of a peak at channel n.

The comparison is made as follows. A quantity

$$DIFF = N_n - 1.96 \sqrt{N_n}$$

is first computed, where N_n is the smoothed count corresponding to channel n. From this quantity is subtracted the average value of the counts in channels $n - \Delta n$ and $n + \Delta n$. If the result is positive, a photopeak located at or near channel n is indicated. The theoretical basis for this procedure is given by Lapp and Andrews (1956). The value of Δn varies with n because of the way in which the peak width varies with its location in the spectrum. The programme user must specify initial (IHWID) and final (LHWID) values for Δn. Intermediate values are calculated automatically by the programme.

At this stage the computer will have identified the location of photopeaks as being in a particular channel n. The true position for the photopeak is now found more accurately by fitting a Gaussian distribution to the spectrum in the region of channel n. Boekelheide (1960) first proposed respresenting photopeaks by a Gaussian function but his procedure requires knowledge of the area under the Gaussian curve. As this is not readily available with the required accuracy, the procedure suggested by Zimmerman (1961) was used. This method requires that the quantity

$$Q(n) = \frac{N_{n-1}}{N_{n+1}} = \exp\left[2(n-n_o)/\delta^2 \right]$$

be computed where N_{n-1}, N_{n+1} are the counts in channels n-1 and n+1 respectively and n_o is the true centre of the Gaussian. The value of n_o is readily obtained since

$$\ell nQ = 2(n-n_o)/\delta^2 \quad ,$$

and the plot of ℓnQ versus n is zero for $n = n_o$. A least mean squares method is then applied by the computer using the subroutine GAUFIT.

This procedure, rather than the simpler alternative of locating the centre of the photopeak at the channel containing the maximum count, is justified by the excellent time stability and linearity (see below) of the spectrometer. Table 1 shows the photopeak positions for four photopeaks determined at approximately hourly intervals throughout one day. The spectral drift is usually less than ± 3/10ths of a channel throughout the course of a day.

Calibration data are treated somewhat differently. Calibration data are identified at the input stage and searched for photopeaks. The number of photopeaks located is then compared for correspondence with the number expected (provided as further input after the deck of cards containing the calibration data have been processed). Next the photon energies are read. If correspondence between the numbers of photopeaks has been achieved, the photon energies are arranged in ascending order and a least mean squares fit performed to evaluate the coefficients b_o, b_1 in the equation

$$E_\gamma = b_o + b_1 n \quad ,$$

where E_γ is the gamma ray energy corresponding to channel n. Values of b_o and b_1 obtained previously may also be read in directly.

If values of b_o and b_1 have been calculated or read in, the programme recognises data not used for spectrometer calibration and will evaluate the energy corresponding to the located photopeak. A linear energy calibration is used rather than a more complicated fit (such as polynomial) as the spectrometer is linear to better than 0.1 per cent.

One other feature is provided in the programme. Relative peak heights are calculated. These indicate relative importance and enable the programme user to concentrate attention on major rather than minor peaks when attempting photopeak assignments. Peak heights are evaluated by averaging the counts in channels $n - \Delta n$ and $n + \Delta n$ and subtracting this average from the count in channel n. The peak heights are then examined to find the maximum and relative peak heights calculated as fractions of this maximum.

The complete output from a set of cards consisting of calibration data and analytical data comprises:

(i) The location of the calibration photopeaks.

(ii) The number of calibration photopeaks detected.

(iii) The values of b_o and b_1.

(iv) The original experimental data sorted out with initial channel identification.

(v) The smoothed experimental data sorted and printed with initial channel identification.

(iv) The number of photopeaks detected, their position, relative height and corresponding gamma photon energy.

4. EXPERIMENTAL VERIFICATION

In order to run the programme, the user must provide the following parameters:

WIDTH - the number of channels to the right and left of a suspected photopeak it is desired to examine in more detail.

FWIDTH - the number of values for $D > 0$ and $D < 0$ to be exceeded between channels n to n-WIDTH and n to n+WIDTH respectively.

IHWID - the initial value for the half-width of the photopeak.

LHWID - the final value for the half-width of the photopeak.

As a result of test runs on many nuclides and their mixtures the following values were found to be very satisfactory using 1024 channels and 7-point cubic smoothing:

WIDTH = 4, FWIDTH = 2, IHWID = 4, LHWID = 6.

With these values no peaks were missed and very few extraneous peaks detected.

Table 2 shows the results of six different examinations of a specimen of supposedly pure radium-226. For this work, the gain of the analyser was set so as to exclude low energy gamma photons. Figures 1 to 5 show a typical spectrum drawn so as to reveal the existence of the photopeaks. The value of the computer programme in locating small peaks in the presence of very large ones is clearly demonstrated by these diagrams, and the output from one analysis is given in Table 5.

Several other points emerge from the data in Table 2. A total of 32 photon energies was indicated and Table 3 shows the frequency with which the photon energies occur. When the sample was counted for longer periods (800 minutes) all of these peaks were clearly visible.

The reproducibility of the energy measurements is particulary good, being typically ± 1 keV. In Table 4 an attempt has been made to assign the most common gamma ray energies to the known members of the radium-226 decay chain (Figure 6).

Complete assignment on the basis of the available data is not possible.

5. THE COMPUTER PROGRAMME

The programme is in Level H FORTRAN IV and is listed as an appendix. The first data input is a title card. The second entry specifies the number of data points per set, the number of points used for smoothing and a code number (CAL) which identifies the type of data.

0 corresponds to data to be processed for photopeaks.

1 corresponds to data to be processed in order to calibrate the spectrometer.

3 indicates that calibration data already exist on cards and are to be read in directly for subsequent use.

The third card specifies the initial channel around which smoothing is to occur. This number must be at least one greater than the number of points used for convolution. Proper selection of this point enables the experimenter to reject the low energy region of the spectrum if necessary.

The fourth card specifies the other input parameters WIDTH, FWIDTH, IHWID, and LHWID required to process a spectrum.

A typical time for computation, including printed output for a set of calibration and unknown data, is 60 seconds.

Photopeaks which occur close to the last channel may be missed if the position of the photopeaks (channel n) is such that

$$n + NOPTS > \text{the number of channels.}$$

Insufficient data will, in this case, cause the programme to miss such peaks (as in Figure 5).

6. CONCLUSIONS

A FORTRAN IV (Level H) programme has been developed for processing the output of high resolution multichannel scintillation spectromers. The programme evaluates the location of photopeaks, peak energy and relative intensity. It is particularly useful for qualitative radiochemical studies.

7. REFERENCES

Blackburn, J. A. (1965). - Computer program for multicomponent spectrum analysis using a least squares method. Analytical Chemistry 37 : 1000-1003.

Boekelheide, J. F. (1960). - Two uses of probability paper in scintillation gamma ray spectroscopy. Rev. Sci. Inst. 31 : 1001-1002.

Drew, D. D., Fite, L. E. and Wainerdi, R. E. (1962). - Applications of computers to Nuclear and Radiochemistry. NAS-NS 3107, p.237.

Kuykendall, W. E. and Wainerdi, R. E. (1960). - An investigation of automated activation analysis. International Atomic Energy Agency Conference on 'The uses of radioisotopes in the physical sciences and industry'. Copenhagen, Denmark.

Lapp, R. E. and Andrews, H. L. (1956). - Nuclear Radiation Physics. Prentice-Hall, Inc., Englewood Cliffs, N. J., 2nd ed.

Savitzky, A. and Golay, M. E. J. (1964). - Smoothing and differentiation of data by simplified least squares procedures. Analytical Chemistry. 36 (8): 1627-1639.

Yule, H. P. (1966). - Data convolution and peak location, peak area and peak energy measurements in scintillation spectrometry. Analytical Chemistry. 38 (1):103-105.

Zimmerman, W. (1961). - Evalutation of photopeaks in scintillation gamma ray spectroscopy. Rev. Sci. Inst. 32 : 1063-1065.

APPENDIX

Programme Listing

```
//
//LOCATE   JOB '          ,7207',A.R.PALMER
//        EXEC      FORTHCLG
//FORT.SYSIN  DD        *
C      LOCATE - GAMMA RAY PEAK RECOGNITION
       DOUBLE PRECISION SMCO,D1
       COMMON SMCO(1050),D1(1050)
       COMMON A(150),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
       COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
       COMMON CALIB,VAL(200)
       COMMON JHS(150),JHE(150)
       INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
       COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
       COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
       CALIB=0
C      READ IN TITLE
    1  READ(1,51,END=99)TITLE
C      READ IN NUMBER OF DATA POINTS (NO) AND NUMBER OF POINTS USED FOR
C      SMOOTHING (NOPTS) AND A CONSTANT (CAL) WHICH DEFINES THE KIND
C      OF DATA TO FOLLOW
C      CAL=0 CORRESPONDS TO ANALYTICAL DATA TO BE PROCESSED, CAL=1
C      CORRESPONDS TO DATA USED TO CALIBRATE THE SPECTROMETER.   CAL=3
C      INDICATES THAT CALIBRATION PARAMETERS ARE AVAILABLE AND ARE TO
C      BE READ IN NEXT
    2  READ(1,52)NO,NOPTS,CAL
       IF(CAL.EQ.3)GO TO 9
C      READ IN INITIAL DATA POINT.   IC  IS THE FIRST DATA POINT AROUND
C      WHICH SMOOTHING OCCURS AND MUST BE AT LEAST ONE UNIT GREATER
C      THAN THE NUMBER OF POINTS USED FOR DATA SMOOTHING
    3  READ(1,53)IC
       READ(1,55)WIDTH,FWIDTH,IHWID,LHWID
C      READ IN DATA, CO IS THE COUNT IN A PARTICULAR CHANNEL
    6  READ(1,56)(CO(I),I=1,NO)
       IF(CAL.EQ.0)GO TO 7
       CALL SMOOTH
       CALL SEARCH
       CALL SPECAL
       GO TO 1
    7  CALL SORT
       CALL SMOOTH
       CALL SEARCH
       GO TO 1
    9  READ(1,57)B0,B1
       CALIB=1
       GO TO 1
   51  FORMAT(20A4)
```

```
52 FORMAT(I4,5X,I1,9X,I1)
53 FORMAT(I3)
55 FORMAT(2(9X,I1),2(8X,I2))
56 FORMAT(1X,10I7)
57 FORMAT(F6.3,4X,F10.6)
99 STOP
   END
```



```
   SUBROUTINE SORT
   DOUBLE PRECISION SMCO,D1
   COMMON SMCO(1050),D1(1050)
   COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
   COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
   COMMON CALIB,VAL(200)
   COMMON JYS(1050),JNEW(100)
   INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
   COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
   COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
201 WRITE(3,202)
   WRITE(3,203)TITLE
   WRITE(3,204)
   WRITE(3,205)
   NUM=1
   K=0
   L=1
206 M=NUM
   N=M+9
   LL=L*400
   KK=K+50
   IF(NO.LT.N)GO TO 211
   WRITE(3,207)NUM,(CO(I),I=M,N)
   NUM=N
   IF(NUM.EQ.LL)GO TO 210
   IF(NUM.EQ.KK)GO TO 208
   NUM=NUM+1
   GO TO 206
208 WRITE(3,209)
   K=KK
   NUM=NUM+1
   GO TO 206
210 WRITE(3,202)
   WRITE(3,203)TITLE
   WRITE(3,204)
   WRITE(3,205)
   K=KK
   L=L+1
   NUM=NUM+1
   GO TO 206
211 WRITE(3,207)NUM,(CO(I),I=M,NO)
202 FORMAT(1H1)
203 FORMAT(1H0,20A4,/)
204 FORMAT(1H ,10H INITIAL  ,40X,10H RECORDED )
205 FORMAT(1H ,10H CHANNEL  ,40X,10H  COUNT   ,/)
207 FORMAT(1H ,2X,I4,4X,10(4XI6))
209 FORMAT(1H )
299 RETURN
   END
```

```
      SUBROUTINE SMOOTH
C     SMOOTH CONVOLUTES THE ORIGINAL DATA BY MEANS OF A CUBIC LAW.     5
C     7 OR 9 POINTS MAY BE USED FOR THE CONVOLUTION PROCESS.  NOPTS
C     SET EQUAL TO 5,7 OR 9 RESULTS IN 5,7 OR 9 POINT SMOOTHING
C     RESPECTIVELY.   NOPTS SET EQUAL TO 0 WILL RESULT IN DELETION OF
C     ALL SUBSEQUENT SUBROUTINES.
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      IF(NOPTS.EQ.0)GO TO 399
      IF(NOPTS.EQ.5)GO TO 305
      IF(NOPTS.EQ.7)GO TO 307
      IF(NOPTS.EQ.9)GO TO 309
C     INITIALISATION SEGMENT
C     LC EQUALS LAST DATA POINT AROUND WHICH SMOOTHING OCCURS
  305 LC=NO-3
      C0=17.0D0
      C1=12.0D0
      C2=-3.0D0
      C3=0.0D0
      C4=0.0D0
      CNORM=35.0D0
      GO TO 310
  307 LC=NO-4
      C0=7.0D0
      C1=6.0D0
      C2=3.0D0
      C3=-2.0D0
      C4=0.0D0
      CNORM=21.0D0
      GO TO 310
  309 LC=NO-5
      C0=59.0D0
      C1=54.0D0
      C2=39.0D0
      C3=14.0D0
      C4=-21.0D0
      CNORM=231.0D0
  310 DO 311 K=IC,LC
      SMCO(K)=C0*CO(K)+C1*(CO(K+1)+CO(K-1))+C2*(CO(K+2)+CO(K-2))
      SMCO(K)=SMCO(K)+C3*(CO(K+3)+CO(K-3))+C4*(CO(K+4)+CO(K-4))
      SMCO(K)=SMCO(K)/CNORM
      A(K)=SMCO(K)
  311 JYS(K)=A(K)
      IF(CAL.GT.0)GO TO 399
      WRITE(3,350)
      WRITE(3,351)TITLE
      WRITE(3,352)NOPTS
      WRITE(3,353)
      WRITE(3,354)
      NUM=IC
      K=NUM
      L=1
  312 M=NUM
      N=M+9
      LL=L*400+(IC-1)
      KK=K+49
      IF(LC.LT.N)GO TO 315
      WRITE(3,355)NUM,(JYS(I),I=M,N)
      NUM=N
      IF(NUM.EQ.LL)GO TO 314
      IF(NUM.EQ.KK)GO TO 313
      NUM=NUM+1
      GO TO 312
  313 WRITE(3,356)
      K=KK
      NUM=NUM+1
      GO TO 312
  314 WRITE(3,350)
      WRITE(3,351)TITLE
      WRITE(3,352)NOPTS
      WRITE(3,353)
      WRITE(3,354)
      L=L+1
      K=KK
      NUM=NUM+1
      GO TO 312
  315 WRITE(3,355)NUM,(JYS(I),I=M,LC)
  350 FORMAT(1H1)
  351 FORMAT(1H0,20A4,/)
  352 FORMAT(1H ,35X21H DATA SMOOTHED USING ,I1,12H POINT CUBIC,/)
  353 FORMAT(1H ,10H INITIAL  ,40X,10H SMOOTHED )
  354 FORMAT(1H ,10H CHANNEL  ,40X,10H  COUNT   ,/)
  355 FORMAT(1H ,2X,I4,4X,10(4XI6))
  356 FORMAT(1H )
  399 RETURN
      END
```

```
      SUBROUTINE SEARCH
C     SEARCH APPLIES TESTS TO THE SMOOTHED DATA TO LOCATE PHOTOPEAKS
      EXTERNAL GAUFIT,EVAL
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      IF(NOPTS.EQ.0)GO TO 490
      IF(NOPTS.EQ.5)GO TO 405
      IF(NOPTS.EQ.7)GO TO 407
      IF(NOPTS.EQ.9)GO TO 409
C     INITIALISATION SEGMENT
  405 LC=NO-3
      C11=8.0D0
      C12=-1.0D0
      C13=0.0D0
      C14=0.0D0
      C1NORM=12.0D0
      GO TO 410
  407 LC=NO-4
      C11=58.0D0
      C12=67.0D0
      C13=-22.0D0
      C14=0.0D0
      C1NORM=252.0D0
      GO TO 410
  409 LC=NO-5
      C11=129.0D0
      C12=193.0D0
      C13=142.0D0
      C14=-86.0D0
      C1NORM=1188.0D0
  410 DO 415 K=IC,LC
      D1(K)=C11*(CO(K+1)-CO(K-1))+C12*(CO(K+2)-CO(K-2))
      D1(K)=D1(K)+C13*(CO(K+3)-CO(K-3))
      D1(K)=D1(K)+C14*(CO(K+4)-CO(K-4))
  415 D1(K)=D1(K)/C1NORM
C     INITIALISATION SEGMENT
  420 J=IC+WIDTH
      LJ=LC-WIDTH
      NOMAX=0
      I=0
  421 JL=J-WIDTH
      JU=J+WIDTH
      IF(D1(J).LE.0.0D0)GO TO 430
      IF(J.EQ.LJ)GO TO 470
      J=J+1
      GO TO 421
  430 IF(D1(J-1).GT.0.0D0.AND.D1(J+1).LT.0.0D0)GO TO 440
      IF(J.EQ.LJ)GO TO 470
      J=J+1
      GO TO 421
  440 NNEG=0
      L=J+1
  441 IF(D1(L).LE.0.0D0)GO TO 445
      IF(L.EQ.JU)GO TO 1450
      L=L+1
      GO TO 441
  445 NNEG=NNEG+1
      IF(L.EQ.JU)GO TO 1450
      L=L+1
      GO TO 441
 1450 NPOS=0
      L=J-1
 1451 IF(D1(L).GE.0.0D0)GO TO 1455
      IF(L.EQ.JL)GO TO 460
      L=L-1
      GO TO 1451
 1455 NPOS=NPOS+1
      IF(L.EQ.JL)GO TO 460
      L=L-1
      GO TO 1451
  460 IF(NPOS.GT.FWIDTH.AND.NNEG.GT.FWIDTH)GO TO 465
      IF(J.EQ.LJ)GO TO 470
      J=J+1
      GO TO 421
  465 I=I+1
      NOMAX=NOMAX+1
      CNUM(I)=J
      IF(J.EQ.LJ)GO TO 470
      J=J+1
      GO TO 421
  470 IF(NOMAX.EQ.0)GO TO 485
      WRITE(3,450)
      WRITE(3,451)TITLE
      N=1
      I=1
      NEWMAX=0
  475 J=CNUM(I)
      K=(IHWID+J*LHWID/NO)/2
      IL=J-K
      IU=J+K
      IF(IL.LT.IC)GO TO 482
      IF(IU.GT.LC)GO TO 482
      DOUBLE PRECISION DIFF
      DIFF=0.0D0
      DIFF=SMCO(J)-1.960D0*SQRT(SMCO(J))
      DIFF=DIFF-0.50D0*(SMCO(IL)+SMCO(IU))
      IF(DIFF.GT.0.0D0)GO TO 480
      IF(I.EQ.NOMAX)GO TO 487
      I=I+1
      GO TO 475
  480 JNEW(N)=J
```

```
      NEWMAX=NEWMAX+1
      VAL(N)=SMCO(J)-0.5*(SMCO(IL)+SMCO(IU))
      N=N+1
      IF(I.EQ.NOMAX)GO TO 487
      I=I+1
      GO TO 475
  482 WRITE(3,1413)J
      IF(I.EQ.NOMAX)GO TO 487
      I=I+1
      GO TO 475
  485 WRITE(3,450)
      WRITE(3,451)
      WRITE(3,1412)
      GO TO 490
  487 IF(NEWMAX.EQ.0)GO TO 489
      NUMBER =NEWMAX
      DO 488 K=1,NUMBER
  488 CNUM(K)=JNEW(K)
      CALL GAUFIT
      IF(CAL.EQ.1)GO TO 1421
      I=1
      TEMP=VAL(I)
      DO 1452 J=2,NEWMAX
      IF(TEMP.GT.VAL(J))GO TO 1452
      TEMP=VAL(J)
 1452 CONTINUE
      DO 1453 I=1,NEWMAX
 1453 VAL(I)=VAL(I)*100./TEMP
      IF(CALIB.GT.0)CALL EVAL
      IF(CALIB.GT.0)GO TO 1420
      WRITE(3,454)
      WRITE(3,455)(CHAN(I), VAL(I), I=1,NUMBER)
      WRITE(3,1414)NEWMAX
      GO TO 490
 1420 WRITE(3,1415)
      WRITE(3,1416)(CHAN(I),VAL(I),ENVAL(I), I=1,NUMBER)
      WRITE(3,1414)NEWMAX
      GO TO 490
 1421 WRITE(3,452)
      WRITE(3,453)(CHAN(I),I=1,NUMBER)
      WRITE(3,1414)NEWMAX
      GO TO 490
  450 FORMAT(1H1)
  451 FORMAT(1H0,20A4,/)
  452 FORMAT(1H ,' PHOTOPEAKS LOCATED AT CHANNELS ',/)
  453 FORMAT(1H ,2XF8.2)
  454 FORMAT(1H0,'PEAK POSITION',5X,'RELATIVE HEIGHT',/)
  455 FORMAT(1H ,T3,F8.2,T22,F7.2)
 1412 FORMAT(1H ,' NO PHOTOPEAKS DETECTABLE ')
 1413 FORMAT(1H ,' PEAK AT CHANNEL ',I4,' NOT CHECKED FOR STATISTICS '/)
 1414 FORMAT(1H0,' NUMBER OF PHOTOPEAKS DETECTED IS ',I3,/)
 1415 FORMAT(1H0,'PEAK POSITION',5X,'RELATIVE HEIGHT',5X,'ENERGY',/)
 1416 FORMAT(1H ,T3,F8.2,T22,F7.2,T39,F6.3)
  489 WRITE(3,1412)
```



```
  490 RETURN
      END
```

```
      SUBROUTINE SPECAL
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      EXTERNAL ORDER,LSQ,GAUFIT
      CALIB=1
  500 READ(1,1504)NPEAKS
C     NPEAKS IS THE EXPECTED NUMBER OF PHOTOPEAKS CONTAINED IN THE
C     CALIBRATION DATA
      IF(NPEAKS.EQ.NUMBER)GO TO 505
      WRITE(3,1505)
      CALIB=0
  505 READ(1,1508)(EN(I),I=1,NPEAKS)
      IF(CALIB.EQ.0)GO TO 599
      DO 510 I=1,NUMBER
  510 A(I)=EN(I)
      NUM=NUMBER
C     THE INPUT DATA IS ARRANGED IN ASCENDING ORDER OF ENERGY
      CALL ORDER
      DO 515 I=1,NUMBER
      EN(I)=A(I)
      Y(I)=EN(I)
  515 X(I)=CHAN(I)
      CALL LSQ
      VARCAL=VAR
      B0CAL=B0
      B1CAL=B1
      AVXCAL=AVX
  520 WRITE(3,1509)B0CAL,B1CAL
 1504 FORMAT(I2)
 1505 FORMAT(1H ,' NUMBER OF PEAKS FOUND DIFFERS FROM THAT EXPECTED '/)
 1508 FORMAT(F6.3)
 1509 FORMAT(1H ,' ENERGY = ',F6.3,' + ',F9.6,' X CHANNEL NO ',/)
  599 RETURN
      END
```



```
      SUBROUTINE ORDER
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      NUNUM=NUM-1
      DO 602 I=1,NUNUM
      K=I+1
      DO 602 J=K,NUM
      TEMP=A(I)
      IF(A(I).LT.A(J))GO TO 602
      A(I)=A(J)
      A(J)=TEMP
  602 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE LSQ
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      SUMX=0.0D0
      SUMY=0.0D0
      SUMXY=0.0D0
      SUMXSQ=0.0D0
      SUMYSQ=0.0D0
      DO 705 I=1,NUM
      SUMX=SUMX+X(I)
      SUMY=SUMY+Y(I)
      SUMXY=SUMXY+X(I)*Y(I)
      SUMYSQ=SUMYSQ+Y(I)*Y(I)
  705 SUMXSQ=SUMXSQ+X(I)*X(I)
      D=NUM*SUMXSQ-SUMX*SUMX
      B1=(NUM*SUMXY-SUMX*SUMY)/D
      B0=(SUMXSQ*SUMY-SUMX*SUMXY)/D
      VAR=(SUMYSQ-((SUMY*SUMY)/NUM)-((B1*D*B1)/NUM))/(NUM-2)
      AVX=SUMX/NUM
      VARB1=VAR*NUM/D
      VARB0=(VAR/NUM)*(1+(AVX*AVX*NUM*NUM/D))
      RETURN
      END
```

```
      SUBROUTINE GAUFIT
      EXTERNAL LSQ
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB0
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      DO 890 LL=1,NUMBER
      J=CNUM(LL)
      L=0
      N=1
  810 CA=CO(J-N)
      CB=CO(J)
      TEMP=CA/CB
      IF(TEMP.GT.0.30)GO TO 820
      GO TO 825
  820 L=L+1
      IF(L.EQ.3)GO TO 825
      N=N+1
      GO TO 810
  825 M=0
      N=1
  830 CA=CO(J+N)
      CB=CO(J)
      TEMP=CA/CB
      IF(TEMP.GT.0.30)GO TO 835
      GO TO 840
  835 M=M+1
      IF(M.EQ.3)GO TO 840
      N=N+1
      GO TO 830
  840 N=M+L+1
      NUM=N
      JL=J-L
      JU=J+M
      DO 850 K=1,N
      DO 849 JJ=JL,JU
      II=JJ-JL+1
      AA=CO(JJ-1)
      AB=CO(JJ+1)
      A(II)=AA/AB
      A(II)=ALOG(A(II))
  849 X(II)=JJ
  850 Y(K)=A(K)
      CALL LSQ
  890 CHAN(LL)=-B0/B1
      RETURN
      END
```

```
      SUBROUTINE EVAL
      DOUBLE PRECISION SMCO,D1
      COMMON SMCO(1050),D1(1050)
      COMMON A(1050),X(200),Y(200),TITLE(20),B0,B1,AVX,VAR,VARB1,VARB2
      COMMON EN(100),VARCAL,B0CAL,B1CAL,AVXCAL,CHAN(200),ENVAL(200)
      COMMON CALIB,VAL(200)
      COMMON JYS(1050),JNEW(100)
      INTEGER CNUM,CO,WIDTH,FWIDTH,CAL
      COMMON CNUM(200),CO(1050),WIDTH,FWIDTH,CAL
      COMMON NO,IC,NOMAX,IHWID,LHWID,NEWMAX,NOPTS,NUMBER,NUM,NPEAKS
      IF(NOPTS.EQ.0)GO TO 999
  905 DO 910 I=1,NUMBER
  910 ENVAL(I)=B0CAL+B1CAL*CHAN(I)
  999 RETURN
      END
/*
//GO.SYSIN    DD   *
```

TABLE 1

VARIATION OF PHOTOPEAK POSITION WITH TIME

Time of Day (hours)	Channel Number			
1050	329.17	778.77	847.71	887.19
1130	329.15	778.89	847.84	887.16
1230	329.33	779.06	848.13	887.36
1400	329.36	779.17	848.06	887.53
1445	329.26	779.18	848.15	887.46
1530	328.91	779.09	847.81	887.39
1615	329.51	779.45	848.34	887.68
1700	329.48	779.28	848.13	887.48
Mean	329.27	779.12	848.03	887.41

TABLE 2

PHOTON ENERGIES (MeV) DETECTED IN SAMPLE OF RADIUM-226

Day of Experiment						No. of Occurrences
277 (a.m.)	277 (p.m.)	283 (a.m.)	283 (p.m.)	284 (a.m.)	284 (p.m.)	
0.147	0.147	0.147	-	0.147	0.147	5
0.185	0.185	0.184	0.185	0.185	0.185	6
0.240	0.241	0.241	0.241	0.241	0.240	6
0.255	-	0.256	-	0.256	-	3
-	-	0.271	-	-	-	1
0.294	0.294	0.294	0.294	0.294	0.294	6
0.349	0.351	0.350	0.350	0.351	0.350	6
-	0.509	-	-	-	-	1
0.608	0.608	0.608	0.608	0.608	0.608	6
0.665	0.665	0.664	0.664	0.664	0.664	6
0.704	-	-	-	-	0.703	2
-	-	-	0.722	-	-	1
0.742	0.742	0.741	0.742	0.742	0.742	6
0.768	0.768	0.767	0.768	0.768	0.768	6
0.785	0.786	0.785	0.785	0.786	0.786	6
0.807	0.806	0.806	0.805	0.806	0.806	6
0.934	0.934	0.934	0.934	0.934	0.934	6
1.121	1.122	1.120	1.121	1.121	1.121	6
1.156	1.155	1.156	1.155	1.155	1.155	6
1.183	1.183	1.181	1.182	1.183	1.183	6
1.240	1.240	1.239	1.239	1.239	1.239	6
1.282	1.284	1.281	1.282	1.282	1.282	6
1.381	1.381	1.379	1.380	1.380	1.380	6
1.408	1.409	1.407	1.408	1.409	1.407	6
1.428	1.429	1.428	1.428	1.427	1.429	6
1.512	1.512	1.510	1.511	1.511	1.511	6
1.537	1.537	-	-	-	-	2
-	-	-	1.588	-	1.587	2
1.666	1.664	1.663	1.665	1.664	1.664	6
-	-	1.696	-	-	-	1
1.734	1.735	1.732	1.733	1.733	1.734	6
1.769	1.769	1.767	1.768	1.768	1.768	6
Number of photopeaks reported						Mean
27	26	27	25	25	26	26

TABLE 3

EXAMINATION OF SAMPLE OF RADIUM-226

Number of Photon Energies Reported in Range 0.047 to 1.769 MeV	32
Number occurring in 6 out of 6 spectra	23
Number occurring in 5 out of 6 spectra	1
Number occurring in 3 out of 6 spectra	1
Number occurring in 2 out of 6 spectra	3
Number occurring in 1 out of 6 spectra	4

TABLE 4

EXAMINATION OF RADIUM-226

Photon Energy (MeV)	Assignment	Photon Energy (MeV)	Assignment
0.147	-	0.934	Bi^{214}
0.185	Ra^{226}	1.121	Bi^{214}
0.240	Pb^{214}	1.156	-
0.255	Pb^{214}	1.183	-
0.271	-	1.240	-
0.294	Pb^{214}	1.282	-
0.349	Pb^{214}	1.381	Bi^{214}
0.509	-	1.408	-
0.608	Bi^{214}	1.428	-
0.665	-	1.512	Bi^{214}
0.704	-	1.537	-
0.722	-	1.588	-
0.742	-	1.666	-
0.768	Bi^{214}	1.696	-
0.785	-	1.734	-
0.807	-	1.769	Bi^{214}

TABLE 5

TYPICAL PROGRAMME OUTPUT

SPECTROMETER CALIBRATION GE(LI) 1024 CHANNELS 8 PEAKS

PHOTOPEAKS LOCATED AT CHANNELS

138.77
153.97
183.30
198.50
269.46
640.04
696.70
729.19

NUMBER OF PHOTOPEAKS DETECTED IS 8

ENERGY = 0.026 + 0.001793 X CHANNEL NO

RADIUM 226 GE(LI) 1024 CHANNELS FOR CHECK

PEAK POSITION	RELATIVE HEIGHT	ENERGY
67.50	4.71	0.147
88.71	29.17	0.185
119.70	37.99	0.240
149.56	65.49	0.294
180.91	100.00	0.350
324.46	47.42	0.608
356.02	3.15	0.664
377.84	1.28	0.703
399.15	2.04	0.742
413.79	6.91	0.768
423.69	1.70	0.786
435.23	0.98	0.806
506.42	3.22	0.934
610.53	11.66	1.121
629.77	1.39	1.155
645.09	1.29	1.183
676.70	3.98	1.239
700.41	1.27	1.282
755.11	3.08	1.380
770.26	1.72	1.407
782.29	0.61	1.429
828.29	1.76	1.511
870.37	0.46	1.587
913.53	0.92	1.664
952.23	1.36	1.734
971.32	8.00	1.768

NUMBER OF PHOTOPEAKS DETECTED IS 26



FIGURE 1. SPECTRUM OF RADIUM-226



FIGURE 2. SPECTRUM OF RADIUM-226

P1416



FIGURE 3. SPECTRUM OF RADIUM-226

P1416



FIGURE 4. SPECTRUM OF RADIUM-226

P1416



FIGURE 5. SPECTRUM OF RADIUM-226

P1416

FIGURE 6

DECAY CHAIN OF RADIUM-226